
PWM AMPLIFIER INTRODUCTION
The recent availability of high-voltage and high-current 

PWM amplifiers in hybrid packages has attracted the interest 
of many designers who traditionally use linear amplifiers. The 
advantage of PWM amplifiers is obvious: efficiency of 70 to 
97%. High efficiency translates to lower internal power loss, 
smaller heat sinks, and reduced overall physical size. 

To make it easier to design with these amplifiers, a simple 
and versatile generic PWM Spice model lets you check out 
PWM waveforms without the fear of blowing up the amplifiers 
or getting shocked by high voltages. The methodology behind 
generating such a model applies not only to hybrid PWM 
amplifiers, but also to monolithic and discrete PWM ampli-
fiers. The inputs to the model come from the PWM amplifier’s 
data sheet, and you can run the model on any commercial 
Spice program. 

Even though a PWM amplifier offers analog signals in and 
analog signals out, its circuit functionality is entirely different 
from a linear amplifier’s. A PWM amplifier modulates a pulse 
train in the time domain and uses LC filtering to extract the 
analog-signal output. You can use PWM amplifiers to emulate 

linear constant- voltage amplifiers or linear constant-current 
amplifiers, both at much higher levels of efficiency. 

If you’re unfamiliar with how a PWM amplifier works, you’re not 
alone. Just like op amps, PWM amplifiers come in many sizes 
and flavors, some with fancy bells and whistles. Fortunately, 
the amplifiers all operate under the same principle. 

A PWM amplifier converts an analog signal into a pulse train 
of variable duty cycle. The analog input controls the duty cycle 
of the output pulse train, which switches on and off once during 
each cycle. When a high output is necessary, the pulse train 
switches on most of the time and vice versa. 

Figure 1a shows a basic PWM amplifier. Vin is the analog 
input of 1 to 8V dc. AOUT is a pulse train, and BOUT is its 
inverse. The PWM oscillator determines the frequency of the 
pulse train, and some PWM amplifiers allow you to put in 
your own PWM oscillator. As Vin changes from its minimum 
to its maximum value, the duty cycle of AOUT changes from 
0 to 100%, and the duty cycle of BOUT changes from 100 to 
0%. The difference voltage of AOUT–BOUT has the same 
pulse train as AOUT but with double the amplitude of 2xVs 
p-p (Figure 1b). 

If you connect a dc brush-type motor across AOUT and 
BOUT, you can control the motor speed with Vin. When you 
set Vin in the middle of its range, for 50% duty cycle at AOUT 
and BOUT, the motor stands still. With Vin at its maximum, 
the motor turns at maximum rpm; with Vin at its minimum, the 
motor reverses direction of rotation and turns at maximum rpm 
again. You can directly connect AOUT and BOUT to a motor 
because the winding inductance of the motor turns the pulsed 
voltage into a rippled dc current whose magnitude controls 
the motor speed and whose polarity controls the clockwise 
or counterclockwise direction of the motor. As Figure 1a 
indicates, most other applications need LC filters to filter out 
the PWM pulse train to ensure that an analog signal appears 
at the load. 

USE A GENERIC SPICE MODEL 
Figure 2 shows the generic Spice subcircuit model of a 

PWM amplifier. V1 is a ramp of fixed frequency. E1 serves as 
a comparator that converts the PWM ramp as it crosses Vin 

into a variable-dutycycle pulse train (Figure 3). S5, V5, S6, and 
V6 limit the amplitude of the pulse train to ±5V. S1/R1, S2/R2, 
S3/R3, and S4/R4 represent the four MOSFET drivers for which 
R1, R2, R3, and R4 are the respective on-resistances. The 
four MOSFETs always turn on and off in diagonal sets, that is, 
when S1 and S4 are on, S2 and S3 are off and vice versa. The 
inverter X1 provides the diagonal switching control. ISENSE 
A and ISENSE B are current-sensing terminals, usually avail-
able at two output pins for current-feedback control circuitry. 
For open-loop operation or for voltage-feedback control, just 
connect ISENSE A and ISENSE B to ground. 

When an external load connects between AOUT and BOUT, 
current flows from Vs to ground through one of two routes: 
Vs to S1/R1, to an externally connected load between AOUT 
and BOUT, to S4/R4, and then to ground or Vs to S3/R3, to 
the external load, to S2/R2, and finally to ground. The voltage 
across the load actually doubles the Vs voltage. For example, 
when Vs=100V, the voltage across the load is 200V pp. This 
voltage-doubling feature is another advantage PWM ampli-
fiers offer for high-voltage applications. To double voltage 
using linear amplifiers you must use two linear amplifiers in a 
bridge-mode configuration. 

DESIGN EXAMPLE: CONSTANT–CURRENT AMPLIFIER
You commonly use constant-current amplifiers for applica-

tions such as motor-torque control and battery chargers. You 
can use the model and the specifications of a commercial 
PWM amplifier—in this case, the Apex SA50—to design a 
constant-current amplifier (also called a voltage-to-current 
converter). You start out with the following specifications from 
the SA50 data sheet: 

Analog input voltage/output duty cycles: 
Vin=4V; AOUT=0% and BOUT=100% 
Vin=6V; AOUT=50% and BOUT=50% 
Vin=8V; AOUT=100% and BOUT=0% 
switching frequency: 45 kHz. 
MOSFET on-resistance: 0.5Ω total or 0.25Ω each 

The analog input voltage range of 4 to 8V dc and the switching 
frequency of 45 kHz determine the waveform of the PWM ramp 
(Figure 4), which V1 in Figure 2 produces. You can describe 
this waveform as a constant-voltage source in any commercial 
Spice program, such as Intusoft’s Model ICAP/4Rx V8.8.1. You 
enter V1’s parameters as manual-driven inputs, and this Spice 
program automatically generates the following statement for 
V1: V1 12 0 PULSE 4 8 0 11.1E-6 11.1E- 6 1E-12 22.2E-6, 
where “12 0’’ designates the two nodes for V1.

The MOSFET on-resistance of 0.25V determines the 
values of R1, R2, R3, and R4. The addition of Rq=600Ω and 
Vcc=12V model the SA50 amplifier’s quiescent current and 
the low-voltage power supply necessary to power the H-bridge 
drive circuitry. 

Figure 5 shows the complete Spice subcircuit for the SA50. 
This basic SA50 can drive a bidirectional motor for which Vin 
controls the motor speed and direction of rotation. You can 
add LC filters that let you drive other loads. Even when driv-
ing a motor, LC filters next to the amplifier module are useful 
for EMI and EMC purposes. Without filters, the long cables 
to the motor carry high-voltage switching pulses and act as 
antennas. Because the waveform across AOUT and BOUT 
is a pulse train of variable duty cycle and because Vin, the 
analog input signal, controls the pulse train’s duty cycle or 
pulse width, you must first filter the PWM pulse train to extract 
the analog output signal. 

In Figure 5, the load comprises Rload and Lload. L1, C4, L3, 
and C5 form a low pass filter with a cut off frequency (Fc) of 4.5 
kHz to filter out the SA50 amplifier’s 45-kHz PWM pulse train. 
A rule of thumb is to set the LC filter’s corner frequency one 
decade below the PWM frequency. Of course, you can push 
the corner frequency higher by using multiple-pole LC filters. 
The equations to calculate filter LC values are as follows:

Because of the filter’s differential configuration, these equa-
tions include a x0.5 factor for L1 and L3 and a x2 factor for 
C4 and C5. In this example, Rload=16Ω, and Fc=4.5 kHz, so 
L1=L3=400 µH, and C4=C5=3.1 µF. Because the load for this 
example is inductive, adding the matching network of R17 and 
C8 creates a combined load of 16Ω. The equations for R17 
and C8 are as follows:

In this example, Lload=1 mH, and Rload=16Ω, so C8=3.9 
µF, and R17=16Ω. Similarly , if you have a capacitive load, 
you can use a LR matching network to make the combined 
load resistive, for which

Figure 6 shows the frequency response of the filter with and 
without the matching network. Ignoring the feedback circuitry of 
X3 and X4, a 1kHz 3.5V p-p sine wave with 6Vdc offset at Vin 
produces a 120V pp sine wave across the load (Figure 7).

To complete the design of a constant-current amplifier , you 
must have some means of sensing the load current and provide 
feedback control in case of a load change. Ra and Rb are the 
two current-sensing resistors. Op amp X4 and its associated 
components serve two purposes: first, as a difference amplifier 
with a gain of 20 that converts the current difference between 
Ra and Rb into a voltage output of – 0.5A/V and, second, as 
a lowpass filter comprising C1, C2, C6, and C7 that filters the 
ripple currents in Ra and Rb with a corner frequency of 4.5 
kHz. The design equations are as follows:

To minimize power losses, you should choose Ra and Rb 
values of 0.01 to 0.1Ω. In this example, Fc=4.5 kHz, R8=R9=10 
kΩ. To minimize loading effects, these resistors must be much 
greater than R13=R15=100Ω. Substituting these values into 

Equation 7 and Equation 8, C6=C7=0.35 µF, and C1=C2=180 
pF. Choosing R10=200 kΩ, Equation 6 yields a gain of –0.5 
A/V.

X3 is an integrator that compares the error voltage from X4 
with the input voltage Ein and provides the correct input voltage 
for the SA50 amplifier to close the feedback loop. The design 
equations for the integrator are as follows:

You can complete the design by choosing R12=R14=10 kΩ 
and C3=71 nF (Figure 6). 

You can now run the Spice program. The load current wave-
forms (Figure 8) are as expected. Note that there is a small 
error between the Spice output and the expected value. 

For example, with Ein=10V, the expected output current 
should be –5A, but Figure 8 shows –4.8A. This difference is 
because of the loss resulting from the 0.25Ω MOSFET’s on-
resistance. If you set the on-resistance to zero, you get exactly 
–5A. Listing 1 is the complete Spice circuit description for the 
constant-current amplifier. 

LISTING 1–SA50 CONSTANT-CURRENT-AMPLIFIER SPICE CIRCUIT
*#save V(1) V(25) @R4[i] @R4[p] V(3) @R1[i] @R1[p] V(5)
*#save V(29) @R3[i] @R3[p] V(7) V(19) @R2[i] @R2[p] V(9)
*#save @Vs[i] @Vs[p] V(10) @S4[i] @S4[p] V(11) @S2[i] @S2[p]
*#save @S3[i] @S3[p] @S1[i] @S1[p] V(12) @V1[i] @V1[p] V(13)
*#save V(18) @E1[i] @E1[p] @V5[i] @V5[p] V(8) @S6[i] @S6[p]
*#save @V6[i] @V6[p] V(17) @Rload[i] @Rload[p] @Lload[i] @Rb[i] @Rb[p]
*#save @Ra[i] @Ra[p] V(16) V(14) V(20) V(21) @R8[i] @R8[p]
*#save @R9[i] @R9[p] @R10[i] @R10[p] @R11[i] @R11[p] @V3[i] @V3[p]
*#save @C1[i] @C2[i] V(24) V(18) V(26) @R12[i] @R12[p] @VEin[i]
*#save @VEin[p] @C3[i] @R14[i] @R14[p] @S5[i] @S5[p] V(15) V(30)
*#save V(2) @L1[i] @L3[i] @C4[i] @C5[i] @R13[i] @R13[p] @C6[i]
*#save @R15[i] @R15[p] @C7[i] V(23) V(29) V(6) V(4) V(21)
*#save V(22) @V2[i] @V2[p] V(27) @Vcc[i] @Vcc[p] @Rq[i] @Rq[p]
*#save V(28) @R17[i] @R17[p] @C8[i] @Rload[i]
*#VIEW TRAN Y1
*#alias Y1 @Rload[i]
.TRAN 22.2E-9 4000E-6 0 22.2E-8 UIC
.PRINT TRAN Y1
R4 1 25 0.25
R1 3 23 0.25
R3 5 29 0.25
R2 7 19 0.25
Vs 9 0 DC=80
S4 29 1 10 0 _S4_mod
.MODEL _S4_mod SW VT=2.5 RON=1E-9 ROFF=1E9
S2 23 7 11 0 _S2_mod
.MODEL _S2_mod SW VT=2.5 RON=1E-9 ROFF=1E9
S3 9 5 11 0 _S3_mod
.MODEL _S3_mod SW VT=2.5 RON=1E-9 ROFF=1E9
S1 3 9 10 0 _S1_mod
.MODEL _S1_mod SW VT=2.5 RON=1E-9 ROFF=1E9
X1 11 10 INV { }
.SUBCKT INV 1 2
* in out
B1 3 0 V= ~V(1)
RD 3 2 1
CD 2 0 .87NF
.ENDS
V1 12 0 PULSE 4 8 0 11.1E-6 11.1E-6 1E-12 22.2E-6
E1 13 0 12 18 1E9
*#save @E1[i] @E1[p]
L1 23 4 400u
V5 15 0 DC=5
S6 11 8 0 13 _S6_mod
.MODEL _S6_mod SW VT=2.5 RON=1E-9 ROFF=1E9
V6 0 8 DC=5
Rload 17 6 16
V2 0 22 DC=12
Lload 4 17 1m
Rb 25 0 0.1
Ra 19 0 0.1
X4 16 14 20 21 22 PA21 { }
.SUBCKT PA21 1 2 3 4 5
* PINOUT ORDER +IN -IN OUT +V -V
Q1 10 1 8 QI1

Q2 11 2 9 QI2
R3 12 8 7.39E+03
R4 12 9 7.39E+03
I2 12 5 3.61E-05
C1 12 5 2.73E-12
R5 12 5 1.11E+06
R1 4 10 8.85E+03
R2 4 11 8.85E+03
C2 10 11 9.00E-12
I1 4 5 3.70E-02
G1 6 15 11 10 1.13E-04
G2 6 15 12 15 6.36E-09
R6 6 15 1.00E+05
D1 6 15 DD
D2 15 6 DD
C3 6 7 3.00E-11
G3 15 7 15 6 8.85E+00
R7 7 15 1E3
D3 7 16 DD
V1 18 16 1.60E+00
D4 17 7 DD
V2 17 19 1.60E+00
RE1 15 0 0.001
E2 18 0 4 0 1
E3 19 0 5 0 1
R8 7 20 50
C4 20 15 3.08E-09
Q3 19 20 21 QOP
Q4 18 20 22 QON
Q5 4 23 29 QON
Q6 5 24 30 QOP
Q7 25 27 31 QLN
Q8 26 28 31 QLP
R11 21 23 1.70E-01
RCLP 29 31 1.70E-01
RCLN 30 31 1.70E-01
R13 22 24 1.70E-01
D5 23 25 DL
D6 26 24 DL
R9 27 29 1E3
R10 28 30 1E3
I3 18 23 7.92E-03
I4 24 19 7.92E-03
R15 31 3 5.42E-01
RSN 3 34 1
CSN 34 5 0.1E-6
.MODEL DD D(CJO=0.1PF IS=1E-17)
.MODEL DL D(CJO=3PF IS=1E-13)
.MODEL QI1 NPN (BF=6.55E+02 IS=8E-16)
.MODEL QI2 NPN (BF=4.24E+02 IS=8.46E-16)
.MODEL QOP PNP (BF=4.64E+02 IS=1E-14)
.MODEL QON NPN (BF=4.64E+02 IS=1E-14)
.MODEL QLN NPN (BF=100 IS=1E-14)
.MODEL QLP PNP (BF=100 IS=1E-14)
.ENDS

R8 16 30 10K
R9 14 2 10K
R10 14 20 200K
R11 16 0 200K
Vcc 27 0 DC=12
V3 21 0 DC=12
C1 16 0 180p
C2 14 20 180p
Rq 27 0 600
X3 0 24 18 21 22 PA21 { }
R12 24 26 10K
VEin 26 0 DC=10
C3 18 24 71n
S5 15 11 13 0 _S5_mod
.MODEL _S5_mod SW VT=2.5 RON=1E-9 ROFF=1E9
L3 29 6 400u
R14 24 20 10K
C4 4 0 3.1u
C5 0 6 3.1u
R17 28 6 16
R13 30 25 100
C6 30 0 0.35u
C8 4 28 3.9u
R15 2 19 100
C7 2 0 0.35u
.END

CONSTANT–VOLTAGE AMPLIFIER 
In applications such as audio-speaker drivers, motor-speed 

control, and power inverters, you need a constant voltage ampli-
fier. You can use the Apex SA02 to design a high efficiency, 
high-power PWM audio-speaker driver. The SA02 data sheet 
lists the following specifications:

Analog input voltage/output duty cycles:
Vin=1.25V; AOUT=0%, BOUT=100% 
Vin=2.50V; AOUT=50%, BOUT=50% 
Vin=3.75V; AOUT=100%, BOUT=0% 
switching frequency: 250-kHz 
MOSFET on-resistance: 0.42Ω total or 0.21Ω each. 

The LC filter design is similar to that of the constant-current 
amplifier except the LC filter requires no matching network 
because of the 8Ω resistive load (Figure 9a). The SA02 ampli-

fier’s PWM frequency is 250 kHz, so the design sets the LC 
filter’s corner frequency to 25 kHz. The design of the difference 
amplifier (X4) is somewhat different, however. This constant-
voltage amplifier configuration senses the output voltage, not 
the output current. The voltage at AOUT and BOUT is much 
higher than the voltage across the current-sensing resistors 
in the previous example. Instead of boosting the gain, resistor 
dividers lower the sense voltage to levels that a small signal 
amplifier can handle. The integrator’s (X3) time constant is 
faster to provide the frequency response necessary for audio 
applications. The SA02 audio-speaker driver has a –10V/V 
voltage gain and a 10-kHz power bandwidth. Figure 9b shows 
the circuit’s input and output waveforms. Note that it takes 
about 50 µsec for the output’s sine wave to stabilize. 

The SA02 has many bells and whistles, such as thermal 
sensing and external-logic shutdown, that the generic model 
does not implement. A design engineer can easily analyze 
these independent features with a paper and pencil. However, 
this simple yet versatile model makes it easy to model the main 
PWM function when manual analysis of this feed-back-control 
circuit becomes unmanageable.

